NASA Contractor Report 181635

PREDICTIVE WALL ADJUSTMENT STRATEGY FOR TWO-DIMENSIONAL FLEXIBLE WALLED ADAPTIVE WIND TUNNEL - A DETAILED DESCRIPTION OF THE FIRST ONE-STEP METHOD

Stephen W. D. Wolf and Michael J. Goodyer

UNIVERSITY OF SOUTHAMPTON
Southampton, England

Grant NSG-7172
January 1988

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665

(NASA-CR-181635) PREDICTIVE WALL ADJUSTMENT STRATEGY FOR TWO-DIMENSIONAL FLEXIBLE WALLED ADAPTIVE WIND TUNNEL: A DETAILED DESCRIPTION OF THE FIRST ONE-STEP METHOD (Southampton Univ.) 29 p CSCL 01A G3/02

N88-19409

Unclas
0130327

Predictive Wall Adjustment Strategy for Two-Dimensional Flexible Walled Adaptive Wind Tunnel.

A Detailed Description of the First One-Step Method

S.W.D. Wolf* and M.J. Goodyer**

Contents

* NRC Post Doctoral Resident Research Associate, NASA Langley Research Center, USA.

** Reader in Experimental Aerodynamics, Department of Aeronautics and Astronautics, University of Southampton, U.K.

1. Introduction

Following the realisation that a simple iterative strategy for bringing the flexible walls of two-dimensional test sections to streamline contours was too slow for practical use[1], Judd proposed[2,3], developed and placed in service[4] what was, as far as we know, the first Predictive Strategy (sometimes called a *"one-step"* method, but see comments on the use of this phrase in Section 3). This Strategy, built into a tunnel's control system, makes use of measurements at the flexible walls of the test section to predict the magnitudes of the adjustments to their shapes required to eliminate their interferences at the model.

During the following years (1976 to date) the software was further developed[5,6] and extensively used and proved up to transonic speeds[7-11]. The later developments described in Section 4 were in the form of refinements and did not involve any change in the underlying principles.

The Predictive Strategy reduced by 75% or more the number of iterations of wall shapes, and therefore the tunnel run-time overhead attributable to the streamlining process, required to reach satisfactory streamlines. As a matter of policy the Strategy has been used to eliminate, as far as is experimentally possible the top and bottom wall interferences. However it should be noted that as a means for reducing the streamlining run-time overhead there remains the option of compromise in the quality of the streamlining coupled with the application of modest corrections.

Because the Strategy is rapid and well proven, it is felt that it would be useful to give a detailed description of the software for others easily to adopt. The Strategy works well in two-dimensional testing at any set of conditions up to those which result in the airfoil's shock just extending to a streamlined wall (usually this would be the suction surface shock just extending to the nearest wall) in a suitably designed test section[1,8].

The Strategy was first implemented in software associated with the running of the low speed Self Streamlining Wind Tunnel (SSWT) in 1976, then on the fully automated Transonic Self Streamlining Wind Tunnel (TSWT) in 1979, both at the University of Southampton, U.K., where the software is still available for use in routine two-dimensional testing[9,10,11]. The simplifications and approximations in its theoretical formulation were influenced by the limited computing power available to the team in the early days, and the algorithm happened also to have been programmed first in BASIC, both influences still being visible in the software. More recently the software has been installed in a computer which controls the flexible walls of the adaptive walled test section in the 0.3-m Transonic Cryogenic Tunnel at NASA Langley Research Center.

The Strategy utilises the velocity distributions along both sides of each flexible wall sketched in Figure 1. The real-side velocity distributions are calculated from measurements of static pressures along the insides of the walls, while the velocities on the outsides of the walls, generated by the imaginary flowfields, are derived by calculation using data from the preceding run. This preceding run may have been the preceding iteration in a series performed with a particular model, but in fact the wall shapes and corresponding imaginary side velocity distributions derived from any previous run may be set and used[†]. The Strategy makes use of this wall information in predicting new wall contours which will eliminate the combined top and bottom wall interference present during the current run, while simultaneously providing the imaginary-side velocity distributions over the new contours.

Wall loading is the evidence of interference: if the real and imaginary velocities differ at any point along a wall then the wall is loaded at that point (and in general this is so everywhere) and therefore the line followed by the wall is not that of a streamline in the infinite flowfield. The object of the Strategy is to predict the wall movement required to eliminate the loading and therefore the interference. Then the wall will be streamlined.

The procedures of the Strategy are embodied in the FORTRAN subroutine WAS (standing for Wall Adjustment Strategy) which is written in a general form. The following sections of the report begin with a brief description of the essentials of the test section hardware, followed by the underlying aerodynamic theory which forms the basis of the Strategy. The subroutine is then presented as the Appendix, broken down into segments with descriptions of the numerical operations underway in each, with definitions of variables.

Two points should be noted. Firstly, the flexible walls need to be adjusted for constant Mach number when the test section is empty to allow for the growth of boundary layers, giving what is called "*aerodynamically straight*" wall shapes. The shapes are functions of Reynolds and Mach numbers, and are not set with the aid of WAS because in moving the walls in the desired direction the subroutine introduces perturbations into the

[†]The word "*run*" is used here in the context of data gathering: a run is a period during which wall pressures (and perhaps other data) are being gathered.

imaginary flowfields which should not exist at this stage in the use of the tunnel. Secondly, when streamlining around a model, the wall contours which are set must allow for the variations in the displacement thicknesses of their boundary layers which are induced by the model's pressure field.

2. Essentials of test section hardware

The test section comprises a pair of rigid sidewalls which support the model in two-dimensional testing, and top and bottom walls made from a convenient flexible material. The flexible walls are fitted with a number of jacks which allow the shapes to be controlled. The walls are bent by the jacks in single curvature only, are cantilevered at their upstream ends and, for minimum interference from length-truncation effects[3], are relatively long and symmetrically disposed fore and aft of the centre of lift of the model. The spacing of the jacks need not be regular; in fact it is usual practice to pitch the jacks more closely in the region of the model than elsewhere because of the stronger curvature in that region.

The wall streamlining process described here relies on measurements of the positions of the walls at each jacking point together with measurements of the wall centreline static pressures, also at each jacking point. Reference Mach number is derived from reference pressures measured in the usual way at the upstream end of the test section.

3. Basic theory of the strategy

In its basic form Judd's Predictive Wall Adjustment Strategy applies to the case of a single impervious thin wall and a model, both lying in an otherwise undisturbed infinite flowfield. His theory applies to the general case of the unstreamlined wall the shape of which is known together with the velocity distributions along each side. There is no assumption of prior knowledge of the aerodynamic behaviour of the model, neither are model measurements a necessary adjunct to the streamlining process.

The wall is loaded as it does not yet follow the desired line of an unloaded streamline in the infinite flowfield. Manifestations of the loading are the differing pressures and associated velocities on either side, although the latter are used for convenience in this section. The physical presence of the wall and the distribution of velocity difference across it may be replaced by a notional vorticity distribution at the wall. The velocity jump (between that on the real side and that on the imaginary side) is a direct measure of the local strength of the vorticity. The distribution of vorticity has the characteristic that the velocity component induced by it in a direction normal to the wall just cancels the sum of the components from other sources thus preventing through-flow. One other source of normal velocity component is the model.

The situation which has just been described, that is a requirement for the distribution of vorticity to prevent through-flow at a point, is eliminated by changing the slope of the wall at the point so that the vorticity's contribution to through-flow is replaced by a change in the component from the free stream. This is done along the whole wall to remove the vorticity everywhere. The operation will be perfect provided the other sources of normal velocity component remain unchanged.

At streamwise station x along a wall positioned above the model the difference in velocity is represented by local vorticity of strength

$$\Gamma(x) = U(x) - V(x)$$

where $U(x)$ is the real-side velocity distribution (derived from pressure measurements) and $V(x)$ is the imaginary-side (calculated) velocity distribution. A velocity component $\upsilon(\xi)$ normal to the wall at streamwise station ξ is induced by the distribution of vorticity. For small slopes this is approximately given by the integral of the elemental contributions of vorticity at x:

$$\upsilon(\xi) \simeq \frac{1}{2\pi} \int_{-\infty}^{\infty} \frac{\Gamma(x)dx}{(\xi - x)}$$

The slope of the wall is adjusted by the amount which is required for a change in the normal component of the free stream velocity to just oppose that due to the vorticity. This requires (for small values of slope) an increment in slope which is given by the approximation

$$\Delta \frac{dy}{dx}(\xi) = \frac{-\upsilon(\xi)}{U_\infty} \quad or \quad \frac{1}{2\pi U_\infty} \int \frac{\Gamma(x)dx}{(x-\xi)}$$

where U_∞ = free stream velocity. This in turn is integrated leading to the change in wall deflection $\Delta y(\xi)$.

Following the removal of the vorticity there are adjustments to velocity either side of the wall amounting to half of the imbalance existing before movement. This is the method by which the velocity distribution is derived for the imaginary side of the wall shape which is to be set for the next run. The increment in imaginary-side velocity at station x arising from the elimination of the vorticity amounts to $(U(x) - V(x))/2$. Hence the imaginary side velocity for the new shape of wall is

$$V(x) + \left(\frac{U(x) - V(x)}{2} \right)$$

This basic theory appears to offer immediate streamlining, the so-called one-step method. However, a one-step method which does not invoke a knowledge of the model's aerodynamic behaviour would require the behaviour not to change with wall shape, whereas the whole of adaptive-wall work arises because model behaviour is dependent on test-section boundary conditions, in this case wall shape. A further change to the test section flow arises from the second wall which is being streamlined simultaneously. These interaction mechanisms cause the wall's predicted shape not to correspond to the required streamline. The modifications to the strategy to account for these effects are introduced in the next section.

4. Modifications in service

4.1 Coupling and scaling

If there were no other changes in the flowfield then the wall loading could be expected to become zero with the new shape. However there will be a change in the behaviour of the model induced by the movement of the wall, but more importantly the requirement to adjust the opposite wall to bring it also to zero loading will introduce a strong interaction. The simultaneous adjustment of each wall using the above simple algorithm does not lead to convergence of the walls to streamlines. Allowance must be made for what may be regarded as, for long wavelength components of wall movement, a one-dimensional continuity effect, a strong source of coupling. Convergence can be obtained by feeding a proportion of the demanded movement of one wall to the other. The process is now iterative because of the wall-model-wall interactions and, in this form, the software results also in an overshoot. That is, the predictions of wall movement are somewhat exaggerated. The latter is reduced by scaling down the predicted wall movements before accounting for the coupling effect. Empirically determined coupling and scaling factors are used. For each of these modifications to wall shape there are appropriate adjustments to the calculation of imaginary-side velocities.

4.1.1 Scaling

Factoring the distribution of vorticity along a wall by factor SF results in the same factoring of slope, wall deflection and increment in imaginary-side velocity. The scaled imaginary-side velocity V_s at station x along the next wall contour to be set is then given by

$$V_s(x) = V(x) + SF.\left[\frac{U(x) - V(x)}{2}\right]$$

for the top wall and similarly for the bottom.

4.1.2 Coupling

Coupling requires a proportion CPLF of one wall's movement to be implanted in the other. The modification to the shape (and slope) of the wall receiving the implant introduces increments to its imaginary-side velocity distribution. The increment is identical to the increment in velocity on the opposite wall had the opposite wall itself been moved an amount factored by the coupling factor, in just the same way as when scaling. Hence the coupled imaginary side velocity $V_c(x)$ for one wall is given by

$$V_c(x) = V_s(x) + CPLF.\left(V_s(x)_o - U(x)_o\right)$$

where suffix '*o*' denotes velocities over the opposite wall. The adjustments to the imaginary velocities arising from scaling and coupling are carried out simultaneously for both walls. Typical values of the factors for both walls are:

coupling	:	0.35
scaling	:	0.8

4.2 Checking of imaginary-side velocities

A key issue is the accuracy of the imaginary-side velocity predictions, since the choice of wall shapes and the judgment of whether or not they are streamlined depend on the predictions. This issue has been addressed in several ways. Firstly the validity (in terms of introducing errors of acceptably small size) of certain approximations in the theoretical basis of the Strategy has been investigated[3] with the conclusion that the errors are compatible with those arising from other sources, for example experimental error. Therefore the reduction of the computational complexity, inherent in the use of a simplified algorithm, was justified. Further checks on the velocities predicted by the Strategy have included:

- analytic checks using straight and streamlined wall information derived from potential flow theory.

- the use of source-sink representations of wall shapes to compute the imaginary-side velocities (1,5,12).

- high subsonic verification of imaginary-side velocity distributions with a streamline curvature program.

- experimental verification of velocity by building a top wall contour into the bottom wall of an empty test section then re-streamlining the top wall and measuring the bottom wall real-side velocity distribution (1,8).

Each of these has led to the conclusion that the imaginary-side velocities computed by the Strategy are reliable.

4.3 Compressibility

The wall adjustment strategy of Section 3 is based on potential flow theory, but linearised compressible flow corrections were introduced in the following manner to allow testing at high subsonic speeds. The various tunnel pressure measurements, in terms of pressure coefficients, C_{pC} are converted to their equivalent incompressible coefficients C_{pI} using

$$C_{pI} = \beta C_{pC} \quad \textit{where} \quad \beta = \sqrt{1 - M_{\infty}^2}$$

and M_{∞} is the reference Mach number.

Velocities derived from the incompressible pressure coefficients are utilised in the strategy. The predicted wall movements and the imaginary-side potential flow velocities are stored, available if required for futher iterations. Some earlier publications[6,8] contained an error, a factoring of movement demands by β, which is corrected here.

This extension to the Strategy successfully allowed testing up to speeds just giving sonic flow at one of the flexible walls.

Acknowledgement

The work described in this report was undertaken for NASA, supported by them under Grant NSG-7172 and by the British Science and Engineering Research Council.

APPENDIX: SOFTWARE DESCRIPTION

The Predictive Strategy is contained in the subroutine WAS (Wall Adjustment Strategy), the essentials of which are presented in the following abstract. There is no specific reference to any particular test section and the subroutine may be used in the control of any test section as long as it is of the type described in Section 2 and is used in two-dimensional testing.

The following data inputs are required by the subroutine:

i) Variables:

NOCPT	Number of computing points along one flexible wall (Top and bottom walls assumed the same). The computing points are located at each of the streamlining jacks and at two dummy positions upstream and downstream as indicated on Figure 2. The dummy positions are referred to as dummy jacks.
FMACH	Free stream Mach Number M_∞
PSTATIC	Reference Static Pressure, P_{REF}
TWSF,BWSF	Top, bottom wall scaling factors
TWCPLF,BWCPLF	Top, bottom wall coupling factors

ii) Arrays:

XJACK(*)	Longitudinal co-ordinates of the wall computing points. The origin of the wall co-ordinates is the wall anchor point.
TOPWP(*),BOTWP(*)	Static absolute pressures at the wall computing points obtained from measurements along the centrelines. Dummy wall pressures are assumed: equal to P_{ref} at the upstream positions and equal to that at jack N for the downstream positions.

TWVEL(•),BWVEL(•) Normalised velocity perturbations at the computing points in the imaginary flows over the current wall contours. Normalising is relative to free stream velocity.

During the current run, the flexible walls have been set to known contours. By this we mean that the contours and their imaginary-side velocity distributions are known. The data is input from file.

Data output includes the arrays:

TWMOV(•),BWMOV(•) Required movements of top, bottom wall jacks from their current positions for streamlining. The convention is positive upwards for both walls.

TWNVEL(•),BWNVEL(•) Top, bottom wall imaginary-side normalised velocity perturbations at computing points, which will apply to the next contours to be set (using TWMOV, BWMOV).

The software is now broken down into logical segments associated with numerical procedures. In this code the word *'velocity'* represents a velocity normalised by the free stream velocity.

Segment 1

In this segment the real-side wall pressures at computing points, measured during the current run, are converted to pressure coefficients, then to equivalent incompressible coefficients. The velocity differences between real and imaginary flows then lead, after scaling and coupling operations, to the new velocities which will exist on the imaginary sides of the next contours to be set.

Notes on code:

Line 52 BETA = Prandtl Glauert Factor = $\sqrt{1 - M_\infty^2}$

Line 53 Q1 = Dynamic Pressure = $\gamma P_{ref} M_\infty^2/2$

Lines 54-84	Loop calculating velocity data at each wall computing point. [I = 1 to NOCPT]	
Line 55	TCPC	= Measured top wall pressure coefficient.
Line 59	TCPI	= Equivalent incompressible top wall pressure coefficient.
Lines 60-61	TVEL	= Current real-side velocity perturbation on top wall (u/U_∞) at computing point I.
Line 62	TWVDIF(I)	= Velocity imbalance across the top wall $= \Gamma/U_\infty$.
Line 63	TWVSQ(I)	= Top wall real-side velocity squared (used in assessment of streamlining quality, not used in the strategy).
Lines 64-72	Repeat of lines 55-63 for the bottom wall computing point I.	

Lines 76,77 TWNVEL(I),BWNVEL(I) = Top, bottom wall imaginary-side new velocity perturbations with scaling.

$$= \quad V_s[I] = v[I] + SF \cdot \left[\frac{U[I] - V[I]}{2} \right]$$

Line 81 TNVEL = Working store for TWNVEL(I).

Lines 82,83 Adjustment of top, bottom wall velocities for coupling. TWNVEL(I) and BWNVEL(I) now represent the velocity perturbations which will exist on the imaginary-sides of the contours to be called by TWMOV( ) and BWMOV( ) respectively in Segment 4.

$$\text{TWNVEL(I), BWNVEL(I)} = V_c[I]$$

$$= V_s[I] + \text{CPLF}(V_s[I]_0 - U[I]_0)$$

Subroutine WAS Software Listing

Segment 1

```
C
C       COMPUTE THE VELOCITY IMBALANCE/WALL VORTICITY AT EACH
C         WALL COMPUTING POINT AND THE EXTERNAL VELOCITIES
C                FOR THE NEXT PREDICTED WALL CONTOURS
C
        BETA = SQRT(1-(FMACH*FMACH))
        Q1 = 0.7 * PSTATIC * FMACH * FMACH
 10     DO 5 I = 1,NOCPT
        TCPC = (TOPWP(I)-PSTATIC)/Q1
C
C       APPLY PRANDTL-GLAUERT FACTOR TO MEASURED TOP WALL CPS
C
        TCPI = BETA*TCPC
        TVRATIO = SQRT(1-TCPI)
        TVEL = TVRATIO-1
        TWVDIF(I) = TVEL-TWVEL(I)
        TWVSQ(I) = (TVEL+1)*(TVEL+1)
        BCPC = (BOTWP(I)-PSTATIC)/Q1
C
C       APPLY PRANDTL-GLAUERT FACTOR TO MEASURED BOT. WALL CPS
C
        BCPI = BETA*BCPC
        BVRATIO = SQRT(1-BCPI)
        BVEL = BVRATIO-1
        BWVDIF(I) = BWVEL(I)-BVEL
        BWVSQ(I) = (BVEL+1)*(BVEL+1)
C
C       APPLY SCALING FACTORS TO THE EXTERNAL VEL. CALCULATIONS
C
        TWNVEL(I) = TWVEL(I)+(TWSF*TWVDIF(I)/2)
        BWNVEL(I) = BWVEL(I)-(BWSF*BWVDIF(I)/2)
C
C       APPLY COUPLING FACTORS TO THE EXTERNAL VELOCITIES
C
        TNVEL = TWNVEL(I)
        TWNVEL(I) = TWNVEL(I)+(BWCPLF*(BWNVEL(I)-BVEL))
        BWNVEL(I) = BWNVEL(I)+(TWCPLF*(TNVEL-TVEL))
 5      CONTINUE
```

Segments 2 and 3

The velocity component normal to a point on a wall induced by its longitudinal distribution of vorticity is calculated. This velocity component is then used to determine a local change of wall slope. With one exception per wall the points on a wall where the induced velocity is determined lie mid-way between computing points, the so-called mid-jack points. A cubic is fitted through four adjacent vorticity data points beginning at the upstream computing point, and the velocity induced at a mid-jack point by the central patch of vorticity (lying between the second and third of the vorticity data points) is determined analytically. The group to which the cubic is fitted is then moved one computing-point downstream and the process repeated until the last computing point is reached, summing the contributions to induced velocity at the same mid-jack point. In Segment 2 the streamwise coordinates of the mid-jack points are first determined, then the coefficients in the cubic equations for all curve fits. Segment 3 executes the integrations and finally determines the required increments in the wall slopes at all mid-jack points.

Notes on code:

Segment 2

Line 90	XMIDJ(1)	= The longitudinal co-ordinate of just the first mid-jack point is made to coincide with the wall anchor point because this point has zero slope and deflection.
Lines 94-95	XMIDJ(I)	= The longitudinal co-ordinate of the Ith mid-jack point lies mid-way between XJACK(I) and XJACK(I+1). NCPT1 = NOCPT-2.

Lines 97-160 Loop for top and bottom wall calculations (NN = 1 and 2 respectively)

Lines 102-126 Loop to compute coefficients for piecewise cubic curve fit to the wall vorticity (in the form $\Gamma(x)/U_\infty$) at each computing point. NCPT2 = NOCPT-3.

Lines 104-110 Load sets of four XJACK values and BWVDIF or TWVDIF values into arrays X and VEL respectively.

Lines 111-125 A curve is fitted to the four sets of vorticity data in arrays X and VEL and the coefficients of the cubic $ax^3 + bx^2 + cx + d$ computed, where

a = CUBCOE (IL,4)

b = CUBCOE (IL,3)

c = CUBCOE (IL,2)

d = CUBCOE (IL,1)

where IL is the counter for each patch of wall vorticity. The geometry of the curve-fit is illustrated on Figure 3.

Segment 3

Lines 133-159	Loop to integrate the vorticity along each wall at the mid-jack points numbered from 2 to NOCPT-2.
Line 134	Xø = Co-ordinate of mid-jack point for which each integration is made.
Line 137	VELSUM = Total sum of vertical velocity induced by the complete wall vorticity. Initially set to zero.
Lines 138-154	Loop to perform the analytical integration of the vorticity-induced normal velocity at mid-jack point Xø for I = 1 to NOCPT-3, where I is the counter for each patch of wall vorticity. See Figure 3.
Line 139	X1 = Co-ordinate of lower limit of integration = XJACK(I+1).
Line 144	X2 = Co-ordinate of higher limit of integration = XJACK(I+2).
Line 153	VELSUM = Summing operation for vertical velocities induced by each patch of wall vorticity where

$$VELSUM = VELSUM + \int_{x1}^{x2} \left(\frac{COEFF\emptyset + COEFF1.x + COEFF2.x^2 + COEFF3.x^3}{(x - X\emptyset)} \right) dx$$

This is solved analytically using four standard integrals coded in lines 147 to 152. Since Xø ≠ $x1$ or $x2$ the singularity x = Xø is avoided.

Lines 156,158	TSLOPE(J),BSLOPE(J) = VELSUM/2π = top, bottom wall required change in local wall slope.

Subroutine WAS Software Listing

Segment 2

```
C
C     MAKE THE FIRST MID-JACK CO-ORD AT THE WALL ANCHOR POINT
C           TO ENSURE A ZERO WALL SLOPE AT THIS LOCATION
C
      XMIDJ(1) = XJACK(2)
C
C     DETERMINE OTHER MID-JACK CO-ORDS BETWEEN WALL COMPUTING
C                         POINTS
      DO 15 I = 2,NCPT1
      XMIDJ(I) = (XJACK(I)+XJACK(I+1))/2
15    CONTINUE
      DO 25 NN = 1,2
C
C     PIECEWISE CUBIC CURVE FIT TO THE WALL VORTICITY USING
C          SETS OF FOUR COMPUTING POINTS (LABELLED 1,2,3,4)
C
      DO 95 IL = 1,NCPT2
      I = IL - 1
40    DO 35 J = 1,4
      X(J) = XJACK(I+J)
      IF (NN.EQ.1) GO TO 50
      VEL(J) = BWVDIF(I+J)
      GO TO 35
50    VEL(J) = TWVDIF(I+J)
35    CONTINUE
      V0 = (VEL(3)-VEL(2))/(X(3)-X(2))
      V1 = VEL(2)-V0*X(2)
      DIST1 = 1/(X(4)-X(1))
      V2 = (VEL(4)-V0*X(4)-V1)/((X(4)-X(2))*(X(3)-X(4)))
      V3 = (VEL(1)-V0*X(1)-V1)/((X(1)-X(2))*(X(3)-X(1)))
      V4 = DIST1*(V2-V3)
      V5 = V3-V4*X(1)
      DIST2 = X(2) + X(3)
C
C     CALCULATE COEFFS. FOR EACH PIECEWISE CUBIC CURVE FIT
C
      CUBCOE(IL,1) = V1-X(2)*X(3)*V5
      CUBCOE(IL,2) = V0+V5*DIST2-V4*X(2)*X(3)
      CUBCOE(IL,3) = V4*DIST2-V5
      CUBCOE(IL,4) = -V4
95    CONTINUE
```

Subroutine WAS Software Listing

Segment 3

```
C     AT EACH MID-JACK PT.,INTEGRATE THE VORTICITY ALONG EACH
C     WALL TO FIND THE INDUCED VERTICAL VELOCITIES, ASSUMED
C     NORMAL TO THE TOP AND BOTTOM WALLS, WHICH MUST BE
C     CANCELLED BY CHANGES IN THE FREE STREAM COMPONENT
C     CAUSED BY LOCAL ADJUSTMENT OF WALL SLOPE
C
      DO 45 J = 2,NCPT1
      X0 = XMIDJ(J)
      X0SQ = X0*X0
      X0CUB = X0SQ*X0
      VELSUM = 0.0
      DO 55 I = 1,NCPT2
      X1 =XJACK(I+1)
      COEFF0 = CUBCOE(I,1)
      COEFF1 = CUBCOE(I,2)
      COEFF2 = CUBCOE(I,3)
      COEFF3 = CUBCOE(I,4)
      X2 = XJACK(I+2)
      X2SQ = X2 * X2
      X1SQ = X1 * X1
      SUM0 = COEFF0+COEFF1*X0+COEFF2*(X0SQ)+COEFF3*(X0CUB)
      X3 = ABS(X2-X0)/ABS(X1-X0)
      X4 = ALOG(X3)
      SUM1 = (COEFF1+COEFF2*X0+COEFF3*X0SQ)*(X2-X1)
      SUM2 = (COEFF2+COEFF3*X0)*((X2SQ)-(X1SQ))/2
      SUM3 = COEFF3*((X2SQ*X2)-(X1SQ*X1))/3
      VELSUM = VELSUM+SUM0*X4+SUM1+SUM2+SUM3
55    CONTINUE
      IF (NN.EQ.2) GO TO 60
      TSLOPE(J) = VELSUM/6.28319
      GO TO 45
60    BSLOPE(J) = VELSUM/6.28319
45    CONTINUE
25    CONTINUE
```

Segment 4

The increments required in wall slope are available at each mid-jack point. These are now integrated to provide wall movement, the integration beginning at the anchor point which remains fixed with zero slope. The general technique is to fit the quadratic equation $ax^2 + bx + c$ through three adjacent values of wall slope increment (as a function of streamwise position). This quadratic equation is then integrated giving a cubic which passes through the predicted changes of wall positions of each of the three mid-jack points shown on Figure 4. The first three coefficients of the cubic equation $Ax^3 + Bx^2 + Cx + D$ are related to those of the quadratic equation as follows

$$A = \frac{a}{3};\quad B = \frac{b}{2}\quad C = c$$

The integration is performed between the x-limits of the two jacks which are straddled by this group of mid-jack points, giving the relative change of curve (wall) height between the two jacks. The process is repeated step-by-step along the whole test section from the fixed upstream end, giving the required movement (and shape) of the complete wall.

Notes on code

Lines 164,165 Initialise top, bottom wall jack movement integrands.

Lines 169,170 TSLOPE(1),BSLOPE(1) Top, bottom wall slopes at XMIDJ(1) (the wall anchor points) set to zero.

Lines 176-216 Loop calculating the jack movement demands for wall streamlining at computing point (I + 2), namely Jack I, NCPT3 = NOCPT-4.

Lines 187-195 Determination of cubic coefficients for top wall position changes, where

$$A = \frac{a}{3};\quad B = \frac{b}{2};\quad C = c$$

Line 196 TMOV = Required movement of jack I on top wall.

Lines 197-204 Determination of bottom wall coefficients.

Line 205	BVOV	= Required movement of jack I on bottom wall.
Lines 209-215	Scaling , then coupling of top and bottom wall jack movement demands.	

Segment 4

```
C
C     INITIALISE WALL MOVEMENT DEMAND ACCUMULATORS
C
      TMOV = 0.0
      BMOV = 0.0
C
C     SET WALL SLOPES AT THE WALL ANCHOR POINTS EQUAL TO ZERO
C
      TSLOPE(1) = 0.0
      BSLOPE(1) = 0.0
C
C     FIND THE JACK MOVEMENTS REQUIRED FOR WALL STREAMLINING,
C     BY PERFORMING INTEGRATIONS OF PIECEWISE QUADRATIC
C           CURVES FITTED TO SETS OF THREE WALL SLOPES
C
      DO 65 I = 1,NCPT3
      I1 = I+1
      I2 = I+2
      TSGRAD = (TSLOPE(I2)-TSLOPE(I1))/(XMIDJ(I2)-XMIDJ(I1))
      BSGRAD = (BSLOPE(I2)-BSLOPE(I1))/(XMIDJ(I2)-XMIDJ(I1))
      XJ1SQ = XJACK(I1)*XJACK(I1)
      XJ2SQ = XJACK(I2)*XJACK(I2)
      XJ1CUB = XJ1SQ * XJACK(I1)
      XJ2CUB = XJ2SQ * XJACK(I2)
      X1 = XMIDJ(I)-XMIDJ(I1)
      X2 = XMIDJ(I2)-XMIDJ(I)
      P1 = (TSGRAD-(TSLOPE(I)-TSLOPE(I1))/X1)/X2
      P2 = TSGRAD - P1*XMIDJ(I2)
      X3 = XJACK(I2)-XJACK(I1)
C
C     TOP WALL - MOVEMENT DEMAND CUBIC COEFFICIENTS
C
      A = P1/3
      B = (P2-P1*XMIDJ(I1))/2
      C = TSLOPE(I1) - P2*XMIDJ(I1)
      TMOV=TMOV+(A*(XJ2CUB-XJ1CUB))+(B*(XJ2SQ-XJ1SQ))+(C*X3)
      P1 = (BSGRAD-(BSLOPE(I)-BSLOPE(I1))/X1)/X2
      P2 = BSGRAD - P1*XMIDJ(I2)
C
C     BOTTOM WALL - MOVEMENT DEMAND CUBIC COEFFICIENTS
C
      A = P1/3
      B = (P2-P1*XMIDJ(I1))/2
      C = BSLOPE(I1) - P2*XMIDJ(I1)
      BMOV=BMOV+(A*(XJ2CUB-XJ1CUB))+(B*(XJ2SQ-XJ1SQ))+(C*X3)
C
C     SCALE JACK MOVEMENT DEMANDS
C
      STMOV = TWSF * TMOV
      SBMOV = BWSF * BMOV
C
C     COUPLE JACK MOVEMENT DEMANDS
C
      TWMOV(I) = STMOV+(BWCPLF*SBMOV)
      BWMOV(I) = SBMOV+(TWCPLF*STMOV)
 65   CONTINUE
```

References

1. Goodyer, M.J.: The Self Streamlining Wind Tunnel. NASA TM-X-72699, Aug. 1975.

2. Judd, M., Goodyer, M.J. and Wolf, S.W.D.: Application of the Computer for On-Site Definition and Control of Wind Tunnel Shape for Minimum Boundary Interference. Presented at AGARD Specialists' Meeting on Numerical Methods and Windtunnel Testing. June 23-24, 1976. In: AGARD-CP-210.

3. Judd, M., Wolf, S.W.D. and Goodyer, M.J.: Analytical Work in Support of the Design and Operation of Two-Dimensional Self-Streamlining Test Sections - Semi-annual Progress Report, Oct. 1975 - Mar. 1976. NASA CR-145019, July 1976.

4. Wolf, S.W.D. and Goodyer, M.J.: Self Streamlining Wind Tunnel: Low Speed Testing and Transonic Test Section Design - Semi-annual Progress Report, Apr. 1976 - Dec. 1976. NASA CR-145257, Oct. 1977.

5. Wolf, S.W.D. and Goodyer, M.J.: Studies of Self Streamlining Wind Tunnel Real and Imaginary Flows - Progress Report to July 1978. NASA CR-158831, Aug. 1979.

6. Wolf, S.W.D.: Control Software for Two Dimensional Airfoil Tests Using a Self-Streamlining Flexible Walled Transonic Test Section - Semi-annual Progress Report to Feb. 1982. NASA CR-165941, Aug. 1982.

7. Goodyer, M.J. and Wolf, S.W.D.: Development of a Self-Streamlining Flexible Walled Transonic Test Section. AIAA Journal, vol.20, no.2, Feb. 1982, pp.227-234.

8. Wolf, S.W.D.: The Design and Operational Development of Self-Streamlining Two-Dimensional Flexible Walled Test Sections. Ph.D. Thesis, University of Southampton. Also NASA CR-172328, March 1984.

9. Wolf, S.W.D.: Selected Data from a Transonic Flexible Walled Test Section. NASA CR-159360, September 1980.

10. Wolf, S.W.D.: Aerodynamic Data from a Two-Dimensional Cambered Airfoil Section in a Shallow Transonic Flexible Walled Test Section. NASA CR-166005, October 1982.

11. Goodyer, M.J.: Tests on a CAST 7 Two-Dimensional Airfoil in a Self-Streamlining Test Section. NASA CR-172291, January 1984.

12. Goodyer, M.J.: Computation of Imaginary-Side Pressure Distributions over the Flexible Walls of the Test Section Insert for the 0.3-m Transonic Cryogenic Tunnel. NASA Contractor Report 172363, June 1984.



FIG. 1 DIVISION OF THE INFINITE TWO-DIMENSIONAL FLOWFIELD INTO REAL AND IMAGINARY PARTS.



FIG. 2 REPRESENTATION OF A FLEXIBLE WALL IN THE WALL ADJUSTMENT STRATEGY.



Normalised vertical velocity induced at XØ by the patch of vorticity extending from X1 to X2 is

$$\frac{1}{2\pi}\int_{x=X1}^{X2}\frac{(ax^3 + bx^2 + cx + d)}{(x - XØ)}\,dx$$

FIG. 3 PIECEWISE ANALYTICAL TECHNIQUE USED TO INTEGRATE THE VORTICITY-INDUCED UPWASH AT MID-JACK POINT XØ.



O Movements derived from integration of increments in wall slope.

× Interpolated demands for jack movement.

FIG. 4 INTERPOLATION OF THE MOVEMENT DEMANDED OF JACK I RELATIVE TO ADJACENT UPSTREAM JACK I-1

NASA
National Aeronautics and Space Administration

Report Documentation Page

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA CR-181635		

4. Title and Subtitle	5. Report Date
Predictive Wall Adjustment Strategy for Two-Dimensional Flexible Walled Adaptive Wind Tunnel - A Detailed Description of the First One-Step Method	January 1988
	6. Performing Organization Code
7. Author(s)	8. Performing Organization Report No.
Stephen W. D. Wolf and Michael J. Goodyer	AASU Memo 85/12
	10. Work Unit No.
9. Performing Organization Name and Address	505-61-01-02
University of Southampton Department of Aeronautics and Astronautics Southampton, Hants, S09 5NH ENGLAND	11. Contract or Grant No. NSG-7172
	13. Type of Report and Period Covered
12. Sponsoring Agency Name and Address	Contractor Report
National Aeronautics and Space Administration Langley Research Center Hampton, VA 23665-5225	14. Sponsoring Agency Code

15. Supplementary Notes

NASA Langley Research Center Technical Monitor: Edward J. Ray

16. Abstract

A major requirement for any adaptive wall wind tunnel is a rapid procedure for wall adaptation/streamlining. This paper is a detailed description of the first one-step method for predicting the wall shapes for adaptation in the flexible walled test section. It is intended that this description (including a breakdown of the software required) should aid those wishing to use this method.

This predictive strategy is rapid and well proven. The strategy works well in two-dimensional testing at any set of conditions up to those where the local Mach number is just sonic with the walls adapted or near adapted. This paper describes the basic theory of the strategy together with the essentials of the test section hardware necessary for successful use of the strategy. We discuss in service modifications to highlight the extensive validation of this predictive wall adjustment strategy since 1976.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Adaptive walls/Flexible walls One-step method 2-D transonic	Unclassified - Unlimited Subject Category - 02

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of pages	22. Price
Unclassified	Unclassified	28	A03

NASA FORM 1626 OCT 86